
Matériels  nécessaire 
 

• Un Amiga 500 avec 1 Mega de Chip et 512Ko de slow ou l’émulateur WINUAE. 

Un Amiga 1200 est quand même fortement conseillé, ces 2méga de chip ne sont pas négligeable et de plus il fait 

fonctionner HRTmon sans problème. (Il est aussi possible de passer par WinUAE bien sûr) 

• Une carte ACTION REPLAY MKIII si l’on passe par un Amiga réel ou ça ROM Image si on passe par WinUAE. 

• Le jeu Original R-TYPE II ou son image CAPS (SPS 2065) 

• Le logiciel Xcopy-Pro en disquette ou image disk selon configuration H/W. 
• Le logiciel ASM-ONE v1.20 ou supérieur. 

• Le binaire rmcloader.bin du SectorLoader de Rob Northen. (Disponible dans ce tuto sous forme Hexa) 

• Optionnel mais plus que conseillé : L’utilisation de HRTmon soit en passant par un amiga Réel (ex Amiga1200 

ou Amiga 500 avec 1 Mega de chip (et slow désactivable) ou bien sûr, passer par ça ROM Image si on passe par WinUAE 

 

 

 

 

 

 
Général   Info 
Ce tutoriel Français est basé sur le tutoriel original de Alpha One. 
Néanmoins, ce document n'est PAS une traduction mot par mot de celui-ci mais plus une nouvelle version. 

Basé sur l'original mais avec des nouvelles informations et parfois une autre approche de la problématique. 

 
 

 

 

 

 

 

 

 

Bon tuto. 
            Gi@nts 

 

  



 

 

 



 

 

Agencement des disquettes Amiga v1.1 
 

En France : 
 

On utilise des termes comme : piste, bloc, secteur, face… 

Piste : 0 à 79 Certaine disquette pousse jusqu’à 81 voire 82 pistes mais le standard reste quand même 80 pistes (de 0 à 79) 

Face : 0/1 ou 1/2 ou A/B, Dessus ou dessous tout simplement. Sur Amiga nous avons deux faces utilisées sur 99% des jeux. 

 

Chaque piste, pour un format standard ‘AmigaDOS’ est composée de plusieurs bloc ou secteur, en général 11 par face. 
Le terme piste peut désigner l’ensemble d’une piste (les deux ‘side’ du disque), ou uniquement une ‘side’ d’une piste. 

Une piste standard amigados est découpée en plusieurs partie appelé bloc, secteur, sector.  

 

 

 

 
 

 

 

Dans d’autre pays :  

 

On utilisera d’autre terme, comme sector, keys, tracks, cylindre, head… 

 
Le terme track par exemple que l’on aurait vite fait de traduire ‘piste’ ne colle pas forcément à notre description française. 

En général, le terme tracks désigne toujours une position sur la disquette mais elle va de 0 à 159 (soit 160 tracks) 

Le maximum étant 160 et non 80 car on a deux faces bien sûres, en fait, elle correspond à une piste sur une face. 

 

Il peut néanmoins arriver que l’on utilise dans des tuto anglo-saxon le terme tracks dans le sens ‘piste’ en français 

(donc de 0 à 79 et non de 0 à 159). Mais en règle générale, il a plutôt une plage de 0 à 160. 

 

C’est le terme cylindre qui ‘colle’ plus à notre définition française de piste. 

En effet, il est courant d’utiliser le terme cylindre pour désigner une position sur la disquette de 0 à 79. 

 
Le terme sector ou key quant à lui corresponds au terme français bloc ou secteur. 

 



Sur une disquette au format Amigados, nous avons 880ko et nous avons 11 secteurs par face, par piste. 

 

La taille d’une piste ayant une valeur physiquement maximum. 

Le nombre maximum de sector sur une piste dépends assez logiquement de la taille de ses sectors. 

 

Pref…beaucoup de terme qui ne sont pas forcément utilisé dans leur sens propre, le mieux est de lire un tuto 

et de comprendre quel sens l’auteur a voulu leurs donner. 
 

Il existe aussi un autre type d’appellation utilisé par exemple par le logiciel MFM-Warp  de Ferox* 

*C’est un programme qui scan le disque en bas niveau et essaye d’en réaliser une copie. 
 

Track Calcul Résultat Format  utilisé sous MFMWarp 

0 0/2 0 et pair 0.0 

1 1/2 0 et impair 0.1 

2 2/2 1 et pair 1.0 

3 3/2 1 et impair 1.1 

156 156/2 78 et pair 78.0 

157 157/2 78 et impair 78.1 

158 158/2 79 et pair 79.0 

159 159/2 79 et impair 79.1 
 

 

On notera que : 

 
 

Le premier secteur (secteur 0) appelé aussi bootbloc commence sur la lowerSide en piste 00 

et se fini en piste 79 sur le upperside 

 

En tracks c’est le même système sauf que l’on terminera en Track 179 et non 79. 

 

La piste Zero est celle situé le plus à l’extérieure du disque. 
 

Le 1er secteur logique, donc le premier bloc sur la disquette, se trouve piste 0 secteur 0 

Les bloc se suivent physiquement mais ne sont pas forcément ordonnée, on parle aussi d’entrelacement. 

 

Le bloc 11 (si on part de 0 bien sur) n’est pas le 1er secteur de la seconde piste mais le 1er secteur de la face suivante. 

(voir image ci-dessus) 

 

En format Amigados, la taille d’un secteur est de 512 octets 

Ce qui nous donnes comme taille disponible : 512*11 secteurs*80 pistes*2 faces = 901 120 octets soit 880Ko 
Une ‘track’ AmigaDos a une taille de 512 * 11 = 5632 en décimal soit $1600 octets 

 

 

Mise en application sous l’AR : 

 
Il existe deux commandes sous l’AR qui permettent de charger sauver des pistes, à savoir : RT et WT 

Elles fonctionnent pareil. 

L’une permet la lecture, l’autre l’écriture. 
 

#RT alias Read Track. Permet le chargement de donnée située sur la disquette vers la mémoire. 

#la première valeur sera la track de départ [0 à 159] à indiquer en hexa.                   /!\ ne pas confondre avec piste 

 

#La seconde valeur sera le nbr de demi track à copié à partir de là. 

 

#WT alias Write Track. Permet la sauvegarde de donnée située en mémoire vers la disquette. 

 

 
Exemples :  

 
RT 20 1 50000 
Start Track = $20 et taille à lire = 1 

On copiera donc la piste !16 (en décimal) side 0 en mémoire $50000 

 
Oui car 20 est donné en hexa, ce qui nous donnes !32 en décimal mais il indique une track (de 0 à 159) PAS en piste. 
Pour avoir l’équivalent en piste on divisera donc par 2 (car deux faces). 

 

$20/2=$10 = !16 (en décimal donc) et comme il n’y a pas de retenu on est sur la face0. 

 

 
RT 21 2 

Start Track = $21 et taille à lire = 2 

On copiera la piste !16 side 1 et la piste !17 side 0 en mémoire 50000 

 

21 est donné en hexa donc $21 = !33 en décimal. 

33/2 = 16.5 ,donc piste 16 side 1 et comme on continue à lire/copier les donnes (taille à lire =2), on continue la copie. 

On change donc de track car on est déjà sur la face 1 (il existe que 2 faces sur une disquette) 

On arrive donc sur la prochaine track à savoir, piste 17 en side 0 puisque que c’est la première face au niveau structure la side 0. 

  



WinUAE 

 
Pour ceux qui utilisent winUAE pour ces tutoriels (j’imagine, la plupart des personnes), Je vous conseille fortement 

d’activer le son des lecteurs de disquette histoire d’entendre ce que le lecteur effectue comme accès. 

HOST -> SOUND -> FLOPPY DRIVE SOUND EMULATION - > DF0 Built-In 

 

 
 

 
Voir même, pour plus d’information. Par exemple afficher sur qu’elle face l’on se trouve, d’activer :  

Host -> Miscellaneous -> Native on-screen display AND RTG on-screen display 

 

  



Part 1 X-Copy 
 
Comme dans Tous bon hack qui se respecte, on va commencer par essayer de copier le disk original. 

Démarrer votre logiciel de copie préféré, à savoir X-copy Pro 

Prendre le mode DOSCOPY+, insérer une disquette vierge en DF1 et la disquette du jeu original en DF0 

Lancer la copie en mode DOSCOPY+ 

 

 

Cela semble assez clair, la copie ne fonctionnera pas et il semble que l'on est là un format spécifique. 

On notera tout de même que l'ensemble des erreurs est de l'ordre de :  

 

Rappel des codes d'erreur de Xcopy :  

 

1. Less or more than 11 sectors 

2. No sync found 

3. No sync after gap found 

4. Header checksum error 
5. Error in header/format long 

6. Data block checksum error 

7. Long track 

8. Verify error 

 

  



Part 2 Analyse de l’image IPF 

 

FILENAME 2065_RType2_AMIGA.ipf 

TYPE Floppy_Disk 

ENCODER CAPS(V1) 

FILE 2065(V1) 

DISK 0 

TRACK 00-83 

SIDE 0-1 

PLATFORM Amiga 

REVOLUTION 2 

PROTECTION Copylock  [T00.1] 
 

Concernant la piste 00.0 rien de bien spécial : 11 secteurs de 512 bytes, standard AmigaDOS 

Par contre, la piste 00.1 elle a bien 11 zones mais pas définit en reconnu en tant que secteurs. 

 

 
 
Ensuite, toutes les pistes se ressemblent, à savoir pas de secteur standard, un codage MFM et une taille de $6214 au niveau Data Track. 

Et ce, de la piste 01.0 jusqu’à la piste 79.1 

 

 
 

On notera au passage la détection de track taguée ‘copylock’ par notre logiciel d’analyse dans l’image IPF. 

On notera aussi qu’il est fort probable, au vue de l’image de la copy via X-COPY que toute les pistes n’aient pas la même tailles de data. 

 



Part 3 Début d’analyse du chargement du jeu 
 
Insérer la disquette originale du jeu dans le lecteur et allumer votre Amiga. 

Si vous passez par un émulateur comme WinUAE, il est assez facile de connaître la piste en court. 

Sur un amiga réel, à moins d'avoir un lecteur ou un montage spécifique affichant les pistes...il est toujours possible d’utiliser le bruit 

de chargement des pistes du lecteur pour les compter mais bon. 

 

Quoi qu'il en soit, voilà ce que l'on peut noter sur les chargements du jeu. 

 
Pistes  Information 

 

00 -> 1  Micro Chargement, RAS, écran noir 

2  -> 13 Ras, écran noir 

13 -> 16 Image jeu r-type 

57 -> 59 Image copyright #1 

59 -> 60 Image copyright #2 

16 -> 18 Ras 

18 -> 20 Image écran de départ du jeu 

21 -> 26 Démo du jeu L1 qui commence après semble-t-il une phase de décompression 

27 -> 32 Démo du jeu L2 qui commence après semble-t-il une phase de décompression 

33 -> 43 Démo du jeu L3 qui commence après semble-t-il une phase de décompression 

 

 



Part 4 Analyse bootblock 
 
 

Toujours avec la disquette originale du jeu dans le lecteur de l'Amiga, nous allons regarder ce qui se passe dans le bootblock 

Entrer dans votre AR 

 

#RT alias Read Track, permet le chargement de track  

 

Taper : RT 0 1 50000 (puis touche ESC à la seconde erreur si présente) puis D 50000  

 

 

On peut voir une boucle de copie de donnée de l'adresse $5004E vers $78000 et pour finir un RTS 

 

La valeur $7C000 est mise dans A0 et on effectue un BSR 50072 
Puis on met la valeur $500 dans A0 et on effectue un BSR 50076 

suivi d'une modification de A7 et finalement le RTS exécute le code en $500 
 

Il est donc fort probable que le code en $50072 et/ou $50076 charge des données en mémoire $500 

On notera aussi la valeur AE3B9CE3 mis dans A0 à l'adresse $5006A 
C'est une valeur... peu commune et juste avant, le chargement de donnée de $E2C dans A0... 
 

Effectuer un redémarrage de votre Amiga et on attend la fin de la 1er phase de chargement* avant d’entrer dans notre AR 

*piste 00->13, voir tableau plus haut 

 

Taper : D 

#D, alias Désassemble permet de désassembler du code. 
 

 

Comme prévue, le code en court d’exécution est dans la zone $78000 
 



 

Taper : R puis D 500 

#R permet d’afficher tous les registres du 68000 
 

 

Cela ressemble à une petite routine de 'copylock' ce qui valide la tag copylock trouvé dans notre analyse de l’image IPF. 

Peut-être pouvons-nous gagner du temps sur la compréhension de celle-ci en allant directement regarder à l'adresse  
notée préalablement : $E2C (voir code plus haut sur image du bootblock en $50066) 

 
 

Taper :  D E2A 

 

 
Il est fort probable que ce soit le code qui soit exécuté après la routine de 'copylock' 

Le test en $E2A est intéressant car l'on compare la valeur AE3B9CE3 avec D0 

On notera au passage que c'est la même valeur que dans notre bootblock préalablement chargé en $50000 
voir $5006A 
 
Redémarrer votre Amiga et dès la fin de la première phase de chargement (jusqu'à piste 13) entrer dans votre AR 

#A, alias Assemble, Instruction qui va permettre de taper du code assembleur. 

Taper : A E2A puis 
 

^000E2A move.w #$f00,$dff180 

^000E32 bra $e2a 

^000E34  

X 

 

 

Le code continue son petit chemin puis, et assez rapidement, l'écran passe au rouge. 

Il est temps pour nous d'entrer dans notre AR, taper : R 

 

 

On peut donc connaître la valeur de D0, à savoir : AE3B9CE3 

C'est la valeur qu'il s’attend à avoir avant lors du test en $E2A et si cela n'est pas le cas, il effectue un BNE 21EA 
Regardons d'ailleurs ce qui se passe en $21EA 
 
Taper : D 21EA 

 

 

Code assez simple à comprendre, on met le fond d'écran en VERT et on boucle sur ce code 
Cela ressemble à 100% à une procédure de fin de parcourt :) 



Check du disk = bon  on continue 

pas bon  On boucle sur ce code d’écran vert. 

 

Passer ce système de protection est assez facile, il suffit d’écraser la valeur de D0 suivi d'un JUMP vers la fin de la routine de 'copylock' 

 

Pour être sûr qu'il n'existe pas d'autre vérification sur cette valeur, le mieux est recherché en mémoire si elle est présente 

à une autre adresse. 

#F alias Find. Permet de chercher des données hexa ou ascii dans la mémoire. 
Taper : f AE 3B 9C E3 

 

 

$7801E étant très proche de $78000 est à coup sur la 1er copie que nous avons vue juste au-dessus. 

Il semblerait donc qu'il existe pour l'instant 4 autres vérifications au zone mémoire soulignées en rouge dans l'image ci-dessous. 

 

$07801E 

$7FFF0C 

$7FF24 

$7FFBA 

 

Nous reviendrons plus tard sur cette problématique. 



Part 5 Analyse et compréhension de la routine de Trackloading 
 
Continuons l’exécution de notre code 

#G comme GO, démarre le code en $e34 comme demandé ici. 

Taper : G E34 

 

Laisser le jeu se charger un peu, et pendant un accès disque entrer à nouveau dans l’AR, Taper : D 

On se trouve quelque part en mémoire vers $FA00. 

 

Un trackloader standard fonctionne avec le contrôleur de disquette, on va donc chercher si les adresses des I/O disk sont utilisées. 

Petite piqure de rappel :  

 
 

#FA, alias Find addressing opcode, permet de chercher une adresse de code modifiant une zone mémoire. 
Taper : FA BFD000 puis FA BFE001 
 

 
$bfe001   bits 6 /Fire0  Aucun intérêt pour nous dans le cadre du trackloader 

$bfe001  bits 7 /Fire1  Aucun intérêt pour nous dans le cadre du trackloader 

$bfe001  bits 4 /TK0  Intéressant ($F998) 

$bfe001  bits 5 /RDY  Intéressant ($FA38) 

$bfe001  bits 1 /LED  Aucun intérêt pour nous dans le cadre du trackloader 

 

On devrait donc trouver notre trackloader au-dessus de l'adresse $F998 (avant l’instruction sur le bit  /RDY), voir image ci-dessus. 

 
D’ailleurs, en $00F998, on peut voir quelque chose d’intéressant, un accès ‘track’ 

 

Jetons un coup d’œil aux codes avant l’adresse $00F998 

 

Taper : D F998 et ‘remonter le code’ comme le montre l’image ci-dessous et regarder un peu le code. 

On trouve quelque chose d’étrange de $F956 à $F95E. Cela ne ressemble pas à du code normal, d'ailleurs Le LINEF en $F946 valide ceci. 

 

 

 

Address Chip Name

BFD000 ciab pra /DTR /RTS /CD /CTS /DSR SEL POUT BUSY

BFD100 ciab prb /MTR /SEL3 /SEL2 /SEL1 /SEL0 /SIDE DIR /STEP

BFE001 ciab pra /FIR1 /FIR0 /RDY /TK0 /WPRO /CHNG /LED OVL

Description / data bits



C'est plus visible sous le logiciel HRTmon* pour info. (Disponible en ROM pour WinUAE et en disquette pour un Amiga Réel) 

HRTmon rappelons le, se comporte un peu comme l’AR, il permet de d’désassembler du code mémoire et de faire tout plein d’autre chose ;) 

*Bon le hic, c’est qu’utiliser HRTmon sous WinUAE n’est pas un problème (buildin in rom), mais avec un Amiga réel… avec un A500+, OK 

Un Amiga 1200, OK, un Amiga 500… à moins d’avoir 1Mega de CHIP (et que, pas de slow), le soft ne vas pas fonctionner… 

 

Quoi qu’il en soit, voila ce que ca donne avec celui-ci. 

 

 

Il y a forte chance que le code commence en $F962 et même... si on regarde bien, $F95E semble autant suspect qu'en $F956. 

Par contre, le code en $F962 lui, semble normal… Hummmm, on va même quand même vérifier quelque chose.  

 

On va désassembler mais à partir de cette adresse-1, donc : $F962-1 et $F95E-1 
Taper : D F962-1 puis D F95E-1 

 
 

! YES. On retombe bien sur notre code ‘normal’ en $F962 puis $F966 mais on peu voir clairement qu’une instruction nous était ‘cachée’ 

à savoir : $F960 ASL.W #3,D0 

 

 

On retombe bien sur notre code (ligne verte image ci-dessus) mais avec une instruction en plus. 

Et on voie assez clairement que le début du code a de grande chance d’être en $F960 

Donc notre trackloader, commencerait à l’adresse $F960 



Si on cherche en mémoire un branchement vers cette adresse, on trouve bien quelque chose, Taper : D F960 

 

 

Puis, bien sûr, par rapport au résultat, taper : FA F958 

 
Il y a visiblement du monde qui saute en $F958 , ce qui est assez logique car R-type2 possède plusieurs niveaux. 

D'ailleurs, rien qu'au début de ce tuto, on a compté pas mal de chargements. 

 

On en profite donc pour noter les adresses concernées : $13F4, $1D82, $1EDE, $1FD0, $2250, $AA42, $AAE8 

 

Bon, voyons ce trackloader en action dans son fonctionnement normal ! 

Redémarrer votre Amiga, laisser finir la 1er phase de chargement (jusqu'à piste 13) puis entrer dans votre AR 

 

Taper : D F958 histoire d’être sur du code que l’on va remplacer, puis, appuyer sur ESC et taper :  

#a, alias Assemble. Permet d’assembler du code assembleur. 
A F958 

^00F958 BRA F958 

^00F95A 

 

Cela sera notre ‘boucle de pause’ avant l’exécution du Trackloader original. 
Taper X pour retourner à l’exécution normal du code. 

 

Assez vite, le jeu semble ne plus rien faire... Entrer dans votre AR, taper : D 

On est bien dans notre ’boucle de pause’ crée juste au-dessus. 

 

Appuyer sur la touche ESC puis R, Noter les registres puis continuer l’exécution du code vers le trackloader en tapant :  G F960 

 

Assez rapidement, le jeu va à nouveau entrer dans notre boucle en $F958 et ne plus rien faire, Entrez de nouveau dans votre AR 
Noter les registres puis continuer l’exécution du code vers le trackloader en tapant :  G F960 

/!\ Continuer ce mode opératoire jusqu'à la fin du jeu et noter les valeurs des registres dans un tableau. 
 

Mode Invulnérable = Commencer une partie, mètre le jeu en pause puis appuyer sur le bouton gauche de la souris et la touche F1, 

Relâcher la touche et le bouton, un écran vert confirme la manipulation. Appuyez sur P pour retourner au jeu. 

 

 

Petit rappel : Pour connaitre l’adresse de retour après un branchement à une sous routine. Elle est stockée dans A7 qui contient  

l’adresse mémoire ou est écrite l’adresse de retour. 

 

Exemple : 

 



Part 6 Tableau des registres lors de l’appel de la routine de trackloader 
 
Tableau1 

BOOT N/A N/A Screen RTYPE Screen (c) #1 Screen (c) #2 N/A Screen Start 

TRK 0 --> 13 0-->1  13 --> 16 57 --> 59  59 --> 60  16 --> 18  18 --> 20 

D0 N/A 00000001 0000000F 00000010 00000002 00000004 00000003 

D1 N/A 00000001 0000001F 0000001F 0000001F 0000001F 00000078 

D2 N/A 0000FFFF 0000FFFF 0000FFFF 0000FFFF 0000FFFF 0000000E 

D3 N/A 0000FFFF 0000FFFF 0000FFFF 0000FFFF 0000FFFF 0000FFFF 

D4 N/A 55555555 55555555 55555555 55555555 55555555 55555555 

D5 N/A 00000000 00000000 00000000 00000000 00000000 00000000 

D6 N/A AAAAF2ED 0000FFFF 9C71FFFF 3A6EFFFF 399C093D 0000FFFF 

D7 N/A 00005045 0000FFFF 1451FFFF 1044FFFF 114FFFF 0000FFFF 

                

A0 N/A 0004CE58 0004CE58 0004CE58 0004CE58 00053000 00053000 

A1 N/A 00DFF000 000102F6 000102F6 000102F6 0005CF4C 00010378 

A2 N/A 00BFD000 00054658 00054658 00054658 0002FD80 0000B4F6 

A3 N/A 00030188 00056E58 00056E58 00056E58 00037A80 000151F0 

A4 N/A 00001558 00059658 00059658 00059658 0003F780 0003F780 

A5 N/A 00C014B6 0007BA00 0007BA00 0007BA00 00047480 000150B8 

A6 N/A 000148F6 000148F6 000148F6 000148F6 000148F6 000148F6 

A7 N/A 000004F4 000004F4 000004F8 000004F8 000004F8 000004F8 

(SP) N/A 00001EE4 00001EE4 00001EE4 0000AA48 000013FA 00001D88 

 

 

 
 

  

F
IR

E
 

N/A Anim Intro Niveau 1 Niveau 2 Niveau 3 Niveau 4 Niveau 5 Niveau 6 

TRK N/A 18 --> 19 19 --> 26 26 --> 32 32 --> 43 43 -->48 48 --> 52 52 --> 57 

D0 N/A 00000000 00000005 00000006 00000007 00000008 00000009 0000000A 

D1 N/A 00000000 0000000E 00063918 00063918 00063918 00063918 00063918 

D2 N/A 00000001 00063000 00063000 00063000 00063000 00063000 00063000 

D3 N/A 00000001 00007D00 00000000 00000000 00000000 00000000 00000000 

D4 N/A 00000082 00000028 00000028 00000028 00000028 00000028 00000028 

D5 N/A 0000001F 00008FCA 00000001 00000001 00000001 00000001 00000001 

D6 N/A 82E10020 82E10080 00000030 00000030 00000030 00000030 00000030 

D7 N/A 0000FFFF 0000FFFF 0000FFFF 0000FFFF 0000FFFF 0000FFFF 0000FFFF 

                  

A0 N/A 00DFF000 0004F180 0004F180 0004F180 0004F180 0004F180 0004F180 

A1 N/A 0000EF36 0001039C 0001039C 0001039C 0001039C 0001039C 0001039C 

A2 N/A 000276C8 0005058C 00007f89A 00007f89A 00007f89A 00007f89A 00007f89A 

A3 N/A 00073602 000796E8 00007F93A 00007F93A 00007F93A 00007F93A 00007F93A 

A4 N/A 0000F1A4 0000F1A4 0007F9DA 0007F9DA 0007F9DA 0007F9DA 0007F9DA 

A5 N/A 00009348 00009348 000150C0 000150C0 000150C0 000150C0 000150C0 

A6 N/A 000148F6 000148F6 000148F6 000148F6 000148F6 000148F6 000148F6 

A7 N/A 000004F8 000004F8 00004F8 00004F8 00004F8 00004F8 00004F8 

(SP) N/A 00001E80 0000AAEE 0000AAEE 0000AAEE 0000AAEE 0000AAEE 0000AAEE 

 

 

Avec ce tableau, on va comprendre le fonctionnement du trackloader 

 
  



Si on regarde de nouveau le code du trackloader on peut voir qu'une lecture mémoire en $FB20 est effectuée. 

Taper : D F960 puis M F920 

#M, alias Visualisation mémoire en HEXA/ASCII 

 

 

Cela ressemble à une filetable 

Si on regarde le début du code du trackloader, ça nous donne : 

 
00F960 ASL.W #3,D0  #Décalage de 3 bit vers la gauche de D0 ce qui nous crée au final D0=D0*8 

00F962 LEA FB20(PC),A1 #On met FB20 dans A1, c'est notre Adr. De départ de notre filetable 

00F966 ADDA.W D0,A1  #Ajoute D0 à A1, ce qui va donc nous positionner dans la filetable 

00F968 MOVE.B (A1)+,D0  #On récupère la première valeur du tableau 'TA1' dans D0, ensuite A1=A1+1 

00F96A MOVE.B (A1)+,D1  #On récupère la seconde valeur du tableau 'TA2' dans D1, ensuite A1=A1+1 

00F96C EXT.W D1  #On efface les 2 premiers bytes de D1 

00F96E MOVE.W (A1)+,D2  #On récupère les deux premières valeurs du tableau 'TA3' dans D2, ensuite A1=A1+2 

00F970 MOVE.L (A1)+,-(A7) #On récupère les quatre premières valeurs du tableau 'TA4' dans SP 

00F972 BSR 0000F9B0  #On se branche sur le sous-programme en $F9B0 

00F974 MOVE.L (A7)+,D0  #On restitue le contenue préalablement sauvé 2 lignes au-dessus vers D0 (check?) 

00F976 RTS   #On revient 

 

Ça nous ‘découpe’ la filetable comme ceci : 

 
 
  



Tableau2 

 

 
 

 

TA1 varie de $02 à $76, (max !118 en décimal. 

Possible qu’il indique ‘track’ de départ. (de !0 à !159) 

 

TA2 varie de $01 à $0B, (max !11 en décimal) 

Vue que le format de ces pistes n'est pas standard AMIGADOS et que l'on a des ERROR 1 'Less or more than 11 sectors'  sous X-Copy, 

TA2 pourrait du coup indiquer un nbr de secteurs custom ou quelque chose dans le genre, un nbr de bloc custom par exemple. 

 

TA3 varie de $0002 à $0121, (max !289 en décimal) 

Il indique sûrement d'une taille de donnée. 

Peut-être en nbr de secteur car il est trop petit pour indiquer directement une taille et trop grosse indiquer autre chose. 

 

TA4 au vue de sa taille, on imagine aussi une taille de donnée, mais en quoi... 

 

 

 

 

Ce qui serait intéressante serait dans un premier temps serait de connaître l'adresse de destination mémoire… 

Il est impossible que ce soit la même tout le temps donc elle doit être passé aussi via les registres vus que dans la filetable, on a rien. 

 

Au vue du tableau des registres (tableau 1), il est impossible qu'elle se trouve en D0-D7 car, D0 à D2 sont déjà utilisé (voir code original au dessus) 
et au vue des valeurs contenu dans D3 à D7 cela semble très improbable. 

On va donc commencer nos tests avec A0 

 

Redémarrer votre Amiga et dès la fin de la première phase de chargement (jusqu'à piste 13) entrer dans votre AR 

Taper : A f958 puis 

^00F958 BRA F958 

^00F95A 

 

#X permet le retour au code Amiga en court. 
X 

 

Très vite l'Amiga semble ne plus rien faire. 

 
On va entrer dans l'AR, afficher les registres et remplir la zone mémoire que l'on pense de destination avec la valeur AA 

puis regarder dans la pile l'adresse de retour et poser une boucle à celle-ci. 

 

Comme prévu (déjà noter dans notre relevé des registres précédent lors des chargements, tableau 1), on tombe sur $4CE58 

 

  

D0 D1
TA1 TA2

$00FB20 N/A 02 01 01 21 00 02 43 CC

$00FB28 SCREEN R-TYPE2 1A 03 00 45 00 00 8B C5

$00FB30 N/A 20 01 00 39 00 00 73 8F

$00FB38 Anim Intro 24 0B 00 1C 00 00 38 AF

$00FB40 Screen Start 27 04 00 08 00 00 10 70

$00FB48 Niveau 1 28 01 00 9F 00 01 3E CB

$00FB50 Niveau 2 35 05 00 7F 00 00 FE 22

$00FB58 Niveau 3 40 01 01 0D 00 02 1A 96

$00FB60 Niveau 4 56 07 00 7C 00 00 F8 5A

$00FB68 Niveau 5 61 00 00 56 00 00 AD 35

$00FB70 Niveau 6 68 03 00 76 00 00 ED 9B

$00FB78 N/A 72 02 00 02 00 00 04 00

$00FB80 N/A 72 04 00 02 00 00 04 00

$00FB88 N/A 72 06 00 02 00 00 04 00

$00FB90 N/A 72 08 00 02 00 00 05 75

$00FB98 Screen © #1 72 0B 00 26 00 00 4D 5C

$00FBA0 Screen © #2 76 02 00 27 00 00 4F EF

INFO
D2
TA3

A7
TA4Adr. Filetable



Entrer dans l'AR et Taper : 

#o alias Fill Memory. Permets-le remplisse avec une valeur précise d’une zone mémoire. 

R puis O AA, 4CE58 60000 

M 4F4 (pour afficher l'adresse de retour stockée dans la pile) puis 

A 1EE4 

^001EE4 bra 1ee4  

^001EE6 

 

G f960 

 

 

L'Amiga va charger quelques pistes et assez vite, ne plus rien faire. Entrer dans l'AR et Taper : 

F AA AA AA AA AA AA AA AA AA, 4CE58 (et appuyer sur ESC dès le 1er affichage) 

 

La fin des données est située en $55A58, ce qui nous donne comme taille : $55A58-$4CE58 = $8C00 

Hummm, ça ne ressemble pas à ce que l'on a dans notre filetable... 
 

Si TA4 est la taille de nos donnée chargée par le trackloader, la fin des données devrait se trouver en $4CE58+$8BC5 = $55A1D 

Mais, si on regarde ça de plus près... 

 

Taper : 
M 55A0D 

 

On peut voir qu'en fait, $55A1D est tout à fait probable comme fin de donnée. 
 

 

TA4 serait donc plutôt la taille des données sans extra-Info 
 

D'ailleurs, si on prend pour référence la taille de ce qui a été lue, à savoir : $8C00 

Que l'on divise par TA3 de 'screen RTYPE2' de notre filetable, à savoir $45 
 
$8C00 = !35840 

$45 = !69 mais en fait, si on part de Zero comme 1ère valeur, cela nous donne un total de !70 valeurs possibles. 

Donc : !35840 / !70 = !512 

 

Tiens donc, exactement la taille d'un secteur AMIGADOS. 

Cela reste à vérifier mais il semblerait que TA3 indique la taille des données lue en nombre de block de 512 bytes 
 

 



Part 7 Extra - Taille d'une piste 
Pour mieux ripper l'ensemble des données il est préférable (mais pas obligatoire) de connaître la taille d'une piste. 
Sur une disquette au format AmigaDOS, elle est de $1600 mais quand est t’il sur notre jeux R-Type II ? 

 

Au vue des erreurs affichées lors de la phase de copie avec le logiciel X-copy pro. L'ensemble des pistes semblent non homogène... 

On va quand même vérifier quelques pistes. La 1er piste étant AmigaDOS, on va faire nos tests sur la Seconde et Troisième TRACK 
 

Redémarrer votre Amiga et dès la fin de la première phase de chargement (jusqu'à piste 13), ou plus exactement, 

dès le début de la seconde phase, entrer dans votre AR 
Taper : A F958 puis 

^00F958 BRA F958 

^00F95A 

X 

 

Très vite l'amiga semble ne plus rien faire. 

On va entrer dans l'AR, afficher les registres, regarder dans la pile l'adresse de retour et poser une boucle à celle-ci. 

puis modifier table filetable pour lire le début de la 3eme TRACK 
 

Entrer dans l'AR et Taper : 

R (on note A0 et A7) 

M 4F4 (On affiche le contenue à l'adresse de A7, à savoir l'adresse de retour stockée dans la pile, ici 1EE4) puis, on crée notre boucle. 

A 1EE4 

^001EE4 bra 1ee4  

^001EE6 

 

M FB28 (Comme vue plus haut, c'est l'adresse FB28 qui est adressé en 1er donc on modifie celle-ci) 

:00FB28 03 00 00 01 00 00 02 00         (3eme track, sector 0, 1 secteur, taille 200) 

 
G F960 

 

Très vite l'amiga semble ne plus rien faire. On re entrer dans l'AR, puis 

Taper : M 4CE58 (4CE58 étant l'adresse de destination du trackloader, contenue dans A0 lors de l'appel de la routine) 

 

 

On note, le début de la TRACK03 est : 00 00 1C 8C 30 39 00 00 1C 12 1A B2 00 00 48 E7 

 

On refait la même manipulation mais pour lire le début de la 4eme track donc modification de $fb28 par 
:00FB28 04 00 00 01 00 00 02 00 (Début 4eme track, sector 0 avec une taille de 1 secteur, taille $200 bytes) 
 

On note, le début de la TRACK04 est : 00 40 3D 40 00 CA 3D 41 00 CC 41 F9 00 02 76 00 

 

Maintenant, on va refaire cette manipulation mais en commençant 1 TRACK avant (donc 2 au lieu de 3    et   3 au lieu de 4) 

Mais en lisant le plus de donnée possible d'une piste. 

Sur une disquette AmigaDOS il y a 11 secteurs, on va donc prendre un peu plus pour 'déborder' sur la prochaine TRACK 

 
Refaite donc la manipulation mais avec 

:00FB28 02 00 00 0F 00 00 1E 00 (Début 2eme track, sector 0 avec une taille de 15 secteurs, taille $1E00 bytes) 

 

Puis on recherche en mémoire le début que l'on a noté de la TRACK03 en Tapant : 

F 00 00 1C 8C 30 39 00 00 1C 12 1A B2 00 00 48 E7, 4CE58     (Début le Premier affichage, appuyer sur ESC) 

Cela nous donne $4E616     donc      $4E616 - $4CE58 = $17BE 

La taille de la TRACK02 est donc de $17B0 au lieu des $1600 Standard AmigaDOS 

 

Refaite la manipulation mais avec 

:00FB28 03 00 00 0F 00 00 1E 00 (Début 3eme track, sector 0 avec une taille de 15 secteurs, taille $1E00 bytes) 

F 00 40 3D 40 00 CA 3D 41 00 CC 41 F9 00 02 76 00, 4CE58 (Dès le début du Premier affichage, appuyer sur ESC) 

Cela nous donne $4E658     donc      $4E658 - $4CE58 = $1800 

 

Bingo, pas la même taille. La taille de la TRACK03 est donc de $1800 au lieu des $1600 Standard AmigaDOS 



Part 8 Extra - Taille d'un secteur 
On ne sait jamais... Même si le 'standard' nous appel vers une taille de 512 bytes, il est préalablement de tester. 
 

On va se passer de copie d'écran vue que ce sont les mêmes manipulations que précédemment mais en travaillant sur les secteurs. 

 

Refaite donc les manipulations mais avec 

:00FB28 02 01 00 01 00 00 20 00 (Début 2eme track, sector 1 avec une taille de 1 secteur, taille $200 bytes) 

 
On note, le début du secteur 1 en TRACK02 est : 70 00 72 00 76 00 48 7A 00 0A 23 DF 00 00 00 10 

 

Refaite donc la manipulation mais avec 
:00FB28 02 00 00 02 00 00 40 00 (Début 2eme track, sector 0 avec une taille de 2 secteurs, taille $400 bytes) 

 

Puis on recherche en mémoire le début que l'on a noté du secteur 1 en TRACK02 

F 70 00 72 00 76 00 48 7A 00 0A 23 DF 00 00 00 10, 4CE58  (Début le Premier affichage, appuyer sur ESC) 

 

Cela nous donne $4D058 donc $4D058 - $4CE58 = $200 

La taille d'un secteur en track2 est donc de $200 bytes à savoir 512 bytes en décimal 

 



Part 9 Calcul avant Rip des pistes 
 

Afin de faire tenir notre RIP de pistes sur une disquette AmigaDOS, on se doit de ripper uniquement les données. 

Pour cela nous devons calculer exactement ce que l'on doit copier. 

 

La premier track est de type AmigaDOS (voir Ecran X-copy) on commencera donc le début de notre rip à la piste $2 et secteur $0 

Si on regarde la filetable ont peu remarquer que la plus grosse valeur en TA1 est de $76 avec une taille de donnée en TA4 de $4FEF 
A savoir Screen © #2 

 

On va déjà regarder de quoi à l'air la fin de ces données. 

 

Redémarrer votre Amiga et dès la fin de la première phase de chargement (jusqu'à piste 13) entrer dans votre AR 

Taper : A F958 puis 

^00F958 BRA F958 

^00F95A 

X 

 

Très vite l'amiga semble ne plus rien faire. 

Entrer dans l'AR et Taper : 

R     (On note A7) 

M 4F4  (On affiche le contenue à l'adresse de A7, à savoir l'adresse de retour stockée dans la pile, ici 1EE4) puis, on met en place notre boucle. 

A 1EE4 

^001EE4 bra 1ee4  

^001EE6 

 

#La commande R affiche les registres du 68000 mais il peut aussi servir à positionner des valeurs de ceux-ci. 
R A0 40000 (On positionne A0 à 40000 pour l'adresse de destination) 

 

O BB, 40000 450EF (On remplit la zone mémoire 40000 à 450EF du pattern BB) 

                                    (450EF étant la somme de la longueur dans la filetable de Screen © #2 + $4000 (adr de départ) + delta de $100) 
 

 

M fb28                               (Comme vue plus haut, c'est l'adresse FB28 qui est adressé en 1er donc on modifie celle-ci) 

:00FB28 76 02 00 27 00 00 4F 4E             (Début 76’eme track, sector 2 avec une taille de $27 secteurs, taille $4F4E bytes) 

 
G F960 

 

Très vite l'amiga semble ne plus rien faire. Entrer dans l'AR et Taper : 

F BB BB BB BB BB BB BB BB BB BB BB BB BB BB BB BB BB, 40000 (Dès le début du Premier affichage, appuyer sur ESC) 

 
 

 

 

Fin de donnée à riper ==> 40 49 63 65 21 DE 00 00 00 00 00 04 DA C0 9E C5 8A BB 3A 6E 28 6B 

 



Part 10 Ripage des pistes 
 
Bon, il n'y a plus qu'a comme on dit. 

On a vu que la taille d'une TRACK n'était pas constante et que celle-ci pouvait atteindre une taille de $1800 ce qui, à ma connaissance est le 

maximum que l'on puisse avoir sur Amiga. On va donc partir de cette base pour nos calculs et ajuster si besoin à la fin. 

 

Au vue de ce que l'on sait et de notre filetable, notre début de rip va commencer en TRACK2 et aller jusqu'à la TRACK 76 + 27 secteurs 

$75 – $2 = $73 

$73 * $1800 = $AC800 

$AC800 + $4E4F = $B164F 

 
Le trackloader du jeu fonctionne en taille de secteur donc : $B164F / $200 = $58B 

 

Comme je suis d'une nature méfiante, on va ajouter encore $F à celui-ci ce qui nous donnes : $58B + $F = $59A 

De toute façon, on a noté la fin des données en hexa donc, on pourra ajuster si nécessaire lors de la sauvegarde. 

 

N'oublions pas que le trackloader a aussi besoin d'une taille de donnée donc : $59A * $200 = $B3400 

 

C'est partie ! Je passe la section commentaire qui reste exactement la même que précédemment. 

 
Redémarrer votre Amiga et dès la fin de la première phase de chargement (jusqu'à piste 13), entrer dans l’AR 

Taper : A F958 puis 

#BRA, Instruction du 68000 qui permet de se brancher à l'adresse indiqué, ici une boucle sur nous-même. 
^00F958 BRA F958 

^00F95A 

X 

 
Très vite l'amiga semble ne plus rien faire. Entrer dans l'AR et Taper : 
A 1EE4 

^001EE4 bra 1ee4  

^001EE6 

 

R A0 40000 

M FB28 

:00FB28 02 00 05 9A 00 0B 34 00 

 

O BB, 40000 f3400 

G f960 

 

Bon, pour une raison que j'ignore, la fonction de recherche de chaîne dans la MKIII ne fonctionne pas bien en mémoire haute... -_-' 

On va donc aller voir à la main. 

 
Taper : M F3400 (et remonter jusqu'à apercevoir la vraie fin des données, en l’occurrence, en $F33E0 on a quelque chose) 

 

 

On a donc une des données (qui colle au passage EXACTEMENT à ce que l'on a noté de la fin des DATA du Screen #2) 

en $F33FF (plus exactement en $F33EF mais bon, soyons prudent, on préfère en garder un peu plus que pas assez) 

 
Ce qui nous donnes une taille finale de : $F33FF - $40000 = $B33FF, Fichtre, on n’était pas loin :) 

 

Et $B33FF en décimal nous donnes : 734 207 Bytes soit environ 717 Ko, on devrait pouvoir stocker ça sur une disquette standard Amiga 😊 

 

Insérer dans votre Amiga une disquette vierge et Taper 

#format, Permet de formater une disquette au format AmigaDOS 
#SM, alias SaveMemory permet donc de sauver la zone mémoire de 40000 à F33FF 
format R-type2_1 

Ready to format disk in drive DF0: (y/n)? y 

Disk ok 
 

SM dump, 40000 F33FF 

 

 



Part 11 Création de notre propre loader 
 

On va reprendre exactement ce qui existe dans le tuto original d'AlphaOne à savoir, le sectorLoader de Rob Northen 

Insérer votre disk ASM One 1.20 ou son image adf dans votre Amiga. 
Une fois notre assembleur préféré chargé, remplacer la Disquette d'ASM one par notre disquette de Sauvegarde en DF0 

Celle contenant le fichier dump de sauvegarde. 

 

 

Après avoir alloué environ 800 Ko de Chipmen, Taper le code suivant 

Il est très bien commenté et ne nécessite à mon sens aucun addon d’explication. 

 
; DISKIMAGE_RNC.s 
; #################################################################### 

; ## R-TYPE II (C) ACTIVISON                                        ## 

; ## DISKIMAGE (133 TRACKS) FOR FLASHTRO TUTORIAL.                  ## 

; ## !! USE CHIPMEM FOR THIS FOR WRITEBACK !!                       ## 

; ## -------------------------------------------------------------- ## 

; ## USAGE:                                                         ## 

; ## 1. <A>  ASSEMBLE SOURCECODE                                    ## 

; ## 2. <J>  EXECUTE SOURCE TO MAKE NECCESSARY PATCHES IN DISKIMAGE ## 

; ## 3. <WT> WRITE BACK 133 TRACKS FROM LABEL #DISK_START           ## 

; ## 4. <CC> CALCULATE BOOTBLOCK-CHECKSUM                           ## 

; ## -------------------------------------------------------------- ## 

; #################################################################### 

 

DISKOFFSET_COPYLOCK_CODE    = 2*$1800+$200   

DISKOFFSET_ORIGINAL_LOADER  = 2*$1800+$200+$F960-$500  

DISKOFFSET_FILETABLE     = 2*$1800+$200+$FB20-$500 

 

 ; OVERWRITE OLD COPYLOCK CODE WITH OUR NEW LOADER 

 ; ----------------------------------------------- 

 

 LEA NEW_COPYLOCK_CODE(PC),A0 

 LEA DISK_START+DISKOFFSET_COPYLOCK_CODE,A1 

 MOVE.L #(COPYLOCK_CODE_END-NEW_COPYLOCK_CODE)-1,D0 

COPY_CL: MOVE.B (A0)+,(A1)+ 

 DBF D0,COPY_CL 

 

 ; OVERWRITE FIRST INSTRUCTION OF ORIGINAL LOADER WITH 

 ; A "JMP $500" -> SO OUR LOADER IS EXECUTED! 

 ; --------------------------------------------------- 

 

 LEA DISK_START+DISKOFFSET_ORIGINAL_LOADER,A0 

 MOVE.W #$4EF9,(A0)+   

 MOVE.L #$500,(A0)+ 

 

 ; THE FILETABLE IS LOCATED AT TRACK #2 SECTOR #0 ON DISK 

 ; AND IS LOADED SEPERATELY BY THE ORIGINAL CODE. WE SKIP THIS PART 

 ; AND COPY IT INTO THE RIGHT PLACE HERE ($FB20 IN MEMORY). 

 ; ---------------------------------------------------------------- 

 

 LEA DISK_START+2*$1800,A0   ; TR #2 SEC #0 

 LEA DISK_START+DISKOFFSET_FILETABLE,A1 

 MOVE.L #$90,D0     ; SIZE OF FILETABLE 

COPY_FT: MOVE.B (A0)+,(A1)+ 

 DBF D0,COPY_FT 

 

 MOVEQ #0,D0     ; END OF PATCHING!! 

 RTS 

; ------------------------------------------------------------------------- 



 

; =========================================================================== 

; THE NEW TRACKLOADER  

; ------------------- 

; HERE COMES OUR NEW LOADER THAT  IS WRITTEN TO THE PLACE OF THE 

; ORGINAL COPYLOCK, IN MEMORY AT $500, ON DISK LOCATED ON TRACK #2, SECTOR #1 

; =========================================================================== 

 

 

NEW_COPYLOCK_CODE:      ; <- $500 IN MEMORY ! 

 

 MOVEQ #0,D1     ; HANDLE THE FILES LIKE THE ORIGINAL 

 MOVEQ #0,D2 

 ASL.W #3,D0 

 LEA $FB20,A1 

 MOVE.B (A1,D0.W),D1    ; D1 = TRACKNUM 

 MULS #$1800,D1     ; * TRACKSIZE $1800 

 MOVE.B 1(A1,D0.W),D2    ; D2 = SECTORNUM 

 MULS.W #512,D2     ; * SECTORSIZE $200 

 ADD.L D2,D1     ; GET CORRECT BYTEOFFSET 

 DIVS.W #512,D1     ; AND TURN IT INTO AN SECTOROFFSET 

 MOVE.W 2(A1,D0.W),D2    ; GET FILESIZE IN SECTORS 

 ADDQ.W #1,D2     ;  

 MOVEQ #0,D3    

 MOVE.L 4(A1,D0.W),-(A7)    ; SAVE REAL BYTESIZE OF FILE ON STACK  

 LEA $2FD80,A1     ; MFM POINTER 

AGAIN: MOVEQ #0,D0     ; DRIVENUMBER (DF0:) 

 BSR.W TLOADER     ; CALL RNC TRACKLOADER 

 TST.B D0     ; LOADING OK? 

 BEQ.B LGUT     ; YEAH -> END TLOADER 

BLINKER:BTST #6,$BFE001     ; 

 BNE.B BLINK     ; OTHERWISE MAKE SCREEN FLASH 

 BRA.B AGAIN     ; AND WAIT FOR LEFT MOUSE TO  

BLINK: MOVE.W D0,$DFF180     ; REPEAT PROCEDURE :P 

 SUBQ.W #1,D0     ; 

 BRA.B BLINKER     ; 

LGUT: MOVE.L (A7)+,D0     ; RESTORE BYTESIZE OF FILE 

 RTS      ; FOR FURTHER GAME-USAGE 

  

TLOADER: INCBIN "RNCLOADER.BIN"    ; ROB NORTHEN SECTORLOADER 

 

COPYLOCK_CODE_END: 

 

; ==<DISKIMAGE>======================================================== 

 

DISK_START: 

 

 DC.B "DOS",0     ; TRACK 0 (BOOT) 

 DC.L 0 

 DC.L 0 

 

 MOVE.W #$2,$1C(A1)    ; READTRACK 

 MOVE.L #2*$1800+$200,$2C(A1)   ; DISKPOSITION 

 MOVE.L #$24400,$24(A1)    ; SIZE 

 MOVE.L #$50000,$28(A1)    ; ADDRESS 

 JSR -$1C8(A6)     ; DOIO() 

 

 LEA COPY_C(PC),A0    ; COPY NEXT PART OF THE  

 LEA $78000,A1     ; BOOTCODE TO $78000 

 MOVE.L #(BOOTEND-COPY_C)-1,D0   ; SO THAT WE DONT OVERWRITE 

COPY: MOVE.B (A0)+,(A1)+    ; OUR OWN BOOTCODE WHEN 

 DBF D0,COPY     ; COPYING GAMEDATA TO $500 !! 

 

 MOVE.W #$7FFF,D0     ; KILL DMA/INTERRUPTS 

 MOVE.W D0,$DFF096 

 MOVE.W D0,$DFF09A 

 MOVE.L #$78000,$80.W    ; CONTINUE AT $78000 IN  

 TRAP #0     ; SUPERVISOR MODE 

     

COPY_C: LEA $50000,A0     ; COPY OUR GAMEDATA DOWN 

 LEA $500.W,A1     ; FROM $50000 TO $500 !! 

 MOVE.L #$24400,D0 

COPY_D: MOVE.B (A0)+,(A1)+ 

 MOVE.W D0,$DFF180 

 SUBQ.L #1,D0 

 BNE.B COPY_D 

 

 MOVE.L #$4E714E71,$E86.W    ; NOP OUT LOADING OF FILETABLE 

 MOVE.W #$4E71,$E8A.W    ; 

 

 MOVE.L #$AE3B9CE3,D0    ; MOVE COPYLOCK KEY TO D0 

 MOVE.L D0,$60.W     ; AND TO ALL MEMORYLOCATIONS 

 MOVE.L D0,$E2C.W 

 MOVE.L D0,$7FF0C 

 MOVE.L D0,$7FF24 

 MOVE.L D0,$7FFBA 

 JMP $E2A.W     ; NO NEED FOR A JMP TO $500. 

       ; WE GO DIRECTLY TO $E2A! 

 

BOOTEND:BLK.B $1800-(BOOTEND-DISK_START),0   ; ZEROES REST OF LONGTRACK 0 

 

 BLK.B $1800,0     ; TRACK 1 (EMPTY) 

 

DUMP: INCBIN "DUMP"     ; TRACK 2 - END 

 

 BLK.B 2577,0     ; ZERO TO GET AN EVEN DISKSIZE 

       ; 133 DOS TRACKS IN ALL 

 

DISK_END: 

 

 



Avant toute exécution, sauvegarder le code sur votre disquette de sauvegarde. 

Celle-ci doit maintenant contenir le fichier dump, le fichier DISKIMAGE_RNC.s et bien sur le binaire rncloader.bin du sectorloader 

 

Basculer en mode 'command line' en appuyant sur ESC puis compiler le tout à l'aide de la commande A, 

puis exécuter le code avec la commande J 

 

Insérer ensuite une nouvelle disquette vierge et Taper : 
WT 

RAM PTR>DISK_START 

DISK PTR>0 

LENGTH>133 

 

Sans oublier la commande pour calculer le nouveau checksum du bootblock (sinon cela ne bootera pas) 
CC 

 

 

Une fois le disk crée, redémarrer votre Amiga avec cette même disquette dans le lecteur et… apprécier le jeu 😊 

 

 



Part 12 Binaire du SectorLoader 'rncloader.bin' de Rob Northen 
 
Pour ceux qui n'ont pas réussi à trouver le binaire rncloader.bin de Rob Northen, voilà la version hexa 

 
00000000  48 E7 7F FC 4E 56 FF DC B6 7C 00 03 66 08 61 00  Hç.üNVÿÜ¶|..f.a. 

00000010  00 DC 60 00 00 CE 38 00 02 44 00 03 3D 44 FF DC  .Ü`..Î8..D..=DÿÜ 

00000020  3D 41 FF DE 3D 42 FF E0 3D 43 FF E2 2D 48 FF E4  =AÿÞ=Bÿà=Cÿâ-Hÿä 

00000030  2D 49 FF E8 E4 58 02 40 00 01 52 40 3D 40 FF EC  -IÿèäX.@..R@=@ÿì 

00000040  70 00 36 02 67 72 70 1E D6 41 B6 7C 06 E0 6E 00  p.6.grp.ÖA¶|.àn. 

00000050  00 92 02 81 00 00 FF FF 82 FC 00 0B 0C 6E 00 01  .’....ÿÿ‚ü...n.. 

00000060  FF EC 67 02 D2 41 3D 41 FF EE 48 41 3D 41 FF F0  ÿìg.ÒA=AÿîHA=Aÿð 

00000070  61 00 04 32 30 2E FF F0 72 0B 92 40 B2 6E FF E0  a..20.ÿðr.’@²nÿà 

